ERROR BAR PLOT

PURPOSE
Generates an error bar plot.

DESCRIPTION
An error bar plot is a graphical data analysis technique for showing the error in the dependent variable and, optionally, the independent variable in a standard x-y plot. As in a standard x-y plot, the vertical axis contains a dependent variable while the horizontal axis contains an independent variable. In addition, it contains error bars in the vertical direction and, optionally, the horizontal direction. The error bars can be either symmetric or asymmetric about the point. The number of arguments on the command determines which type of error bars are produced (see the SYNTAX section below).

By using the CHARACTERS and LINES command, the analyst has a great deal of flexibility in formatting the appearance of the error bars. The appearance of the error bars is controlled by the following character and line traces.

Trace 1 = the point (x,y)
Trace 2 = the point (x,y+error)
Trace 3 = the point (x,y-error)
Trace 4 = the point (x-error,y)
Trace 5 = the point (x+error,y)
Trace 6 = the line between (x,y+error) and (x,y-error)
Trace 7 = the line between (x-error,y) and (x+error,y)

The line setting for trace 1 is used if you want the plot points to be connected. The line settings for traces 2 through 5 are usually set to BLANK (some analysts prefer connected lines for traces 2 and 3). Most analysts prefer to use solid lines for traces 6 (connects the vertical errors) and trace 7 (connects the horizontal errors). However, you can use a different line style or leave them blank if you prefer. The character setting for trace 1 is the original data point. Use character traces 2 through 5 if you want the 4 end points to have unique settings. Use character traces 6 and 7 to use the same symbol for matching end points. The program below demonstrates how to set the line and character traces.

SYNTAX 1
ERROR BAR PLOT <y1> <ypos> <SUBSET/EXCEPT/FOR qualification>
where <y1> is the dependent variable;
<ypos> is the error for <y1> in both the positive (up) and the negative (down) direction;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

SYNTAX 2
ERROR BAR PLOT <y1> <ypos> <x1> <SUBSET/EXCEPT/FOR qualification>
where <y1> is the dependent variable;
<ypos> is the error for <y1> in both the positive (up) and the negative (down) direction;
<x1> is the independent variable;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

SYNTAX 3
ERROR BAR PLOT <y1> <ypos> <yneg> <x1> <SUBSET/EXCEPT/FOR qualification>
where <y1> is the dependent variable;
<ypos> is the error for <y1> in the positive (up) direction;
<yneg> is the error for <y1> in the negative (down) direction;
<x1> is the independent variable;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

SYNTAX 4
ERROR BAR PLOT <y1> <ypos> <yneg> <x1> <xpos> <SUBSET/EXCEPT/FOR qualification>
where <y1> is the dependent variable;
<ypos> is the error for <y1> in the positive (up) direction;
<yneg> is the error for <y1> in the negative (down) direction;
<x1> is the independent variable;
<xpos> is the error for <x1> in both the positive (left) and the negative (right) direction;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.
SYNTAX 5
ERROR BAR PLOT <y1> <ypos> <yneg> <x1> <xpos> <xneg> <SUBSET/EXCEPT/FOR qualification>
where <y1> is the dependent variable;
<ypos> is the error for <y1> in the positive (up) direction;
<yneg> is the error for <y1> in the negative (down) direction;
<x1> is the independent variable;
<xpos> is the error for <x1> in the positive (left) direction;
<xneg> is the error for <x1> in the negative (right) direction;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES
ERROR BAR PLOT Y1 YDelta
ERROR BAR PLOT Y1 YDelta X1
ERROR BAR PLOT Y1 YDELPOS YDELNEG X1
ERROR BAR PLOT Y1 YDELPOS YDELNEG X1 XDELTA
ERROR BAR PLOT Y1 YDELPOS YDELNEG X1 XDELPPOS XDELPNEG
ERROR BAR PLOT Y1 YDELTA X1 SUBSET X1 < 10

DEFAULT
None

SYNONYMS
None

RELATED COMMANDS
LINES = Sets the type for plot lines.
CHARACTER = Sets the symbol for plot characters.
PLOT = Generates a data or function plot.

APPLICATIONS
Exploratory Data Analysis, Presentation Graphics

IMPLEMENTATION DATE
88/11
PROGRAM

LET FUNCTION F = X1**2
LET X1 = SEQUENCE -5 1 5
LET Y1 = F; LET N = SIZE Y1
LET DELTA = NORMAL RANDOM NUMBERS FOR I = 1 N
LET DELTA = 2*DELTA
LET YPOS = ABS(DELTA)
LET YNEG = UNIFORM RANDOM NUMBERS FOR I = 1 N
LET YNEG = 3*YNEG
LET XPOS = PATTERN .2 .4 .6 FOR I = 1 N
LET XNEG = PATTERN .4 .8 1.2 FOR I = 1 N

MULTIPLOT 2 2; MULTIPLOT CORNER COORDINATES 0 0 100 100
XLIMITS -5 5; XTIC OFFSET 1 1
YLIMITS 0 30; YTIC OFFSET 2 1
CHARACTER FONT SIMPLEX ALL; CHARACTER CIRCLE - -
CHARACTER SIZE 2.0 3.0 3.0 3.0 3.0 3.0 3.0
CHARACTER FILL ON

TITLE AUTOMATIC
PLOT Y1 VS X1
TITLE SYMMETRIC ERROR BARS
ERROR BAR PLOT Y1 DELTA X1
TITLE ASYMMETRIC ERROR BARS
ERROR BAR PLOT Y1 YPOS YNEG X1
CHARACTER CIRCLE BLANK BLANK BLANK BLANK - |
LINE BLANK BLANK BLANK BLANK BLANK SOLID SOLID
TITLE X AND Y ERROR BARS
ERROR BAR PLOT Y1 YPOS YNEG X1 XNEG XPOS
END OF MULTIPLY